
⭐⭐⭐ Uitdaging ⏱️ 30 minuten

Parkeersensor luxe

De luxe versie van de parkeersensor! Hij piept sneller als je dichterbij komt, en het OLED-

scherm laat de afstand zien in centimeters. Met een balkje dat langer wordt als je dichterbij

komt.

🧰 Wat heb je nodig?

1× Arduino UNO 1× Afstandssensor HC-

SR04

1× Actieve zoemer

1× OLED-schermpje

0,91"

🔌
USB-kabel +

Jumperkabels

🗺️ Aansluitschema

🛠️ Zo bouw je het

1. Sluit de HC-SR04 aan: Trig naar pin 9, Echo naar pin 10, Vcc naar 5V en

Gnd naar GND.

2. Sluit de zoemer aan: S naar pin 8, + naar 5V en − naar GND.

3. Sluit het OLED-scherm aan: SDA naar A4, SCL naar A5, VCC naar 5V en

GND naar GND.

4. Drie onderdelen hebben 5V en GND nodig. Gebruik de + en − strips van je

breadboard.

💻 De code

🧠 Wat doet de code?

meetAfstand()  meet de afstand in centimeters met de HC-SR04.

toonAfstand()  zet het getal groot op het scherm en tekent een balkje met

fillRect() .

De zoemer piept sneller naarmate de afstand kleiner wordt.

🎯 Probeer eens…

Laat "STOP!" op het scherm zien als je dichter dan 5 cm bent

Laat de afstand in meters zien als iets verder dan 1 meter weg is

Voeg de LED-ring toe die van groen naar rood kleurt

🚀 Bouw dit project met Codey: open codey.online/nl/starter-kits/uno/projects/parking-sensor-deluxe en klik op MAAK DIT. Dan staan de code en het

aansluitschema meteen klaar.

Arduino UNO

Afstandssensor HC-SR04 Actieve zoemer

OLED-schermpje 0,91"

// Parkeersensor luxe

// Net als de gewone parkeersensor, maar nu zie je de afstand ook op het 

OLED-scherm.

#include <Wire.h>

#include <Adafruit_GFX.h>

#include <Adafruit_SSD1306.h>

const int TRIG_PIN = 9;     // Trig van de HC-SR04 op pin 9

const int ECHO_PIN = 10;    // Echo van de HC-SR04 op pin 10

const int ZOEMER_PIN = 8;   // S van de zoemer op pin 8

Adafruit_SSD1306 scherm(128, 32, &Wire, -1);  // OLED: SDA = A4, SCL = A5

// Meet de afstand in centimeters

long meetAfstand() {

  digitalWrite(TRIG_PIN, LOW);

  delayMicroseconds(2);

  digitalWrite(TRIG_PIN, HIGH);

  delayMicroseconds(10);

  digitalWrite(TRIG_PIN, LOW);

  long tijd = pulseIn(ECHO_PIN, HIGH, 30000);

  if (tijd == 0) return 999;   // Geen echo: niets in de buurt

  return tijd / 58;

}

// Zet de afstand groot op het scherm, met een balkje eronder

void toonAfstand(long afstand) {

  scherm.clearDisplay();

  scherm.setTextSize(2);

  scherm.setCursor(0, 0);

  if (afstand >= 999) {

    scherm.print(F("---"));

  } else {

    scherm.print(afstand);

    scherm.print(F(" cm"));

  }

  // Balkje: hoe dichterbij, hoe langer

  int lengte = map(constrain(afstand, 0, 100), 0, 100, 128, 0);

  scherm.fillRect(0, 24, lengte, 8, SSD1306_WHITE);

  scherm.display();

}

void setup() {

  pinMode(TRIG_PIN, OUTPUT);

  pinMode(ECHO_PIN, INPUT);

  pinMode(ZOEMER_PIN, OUTPUT);

  scherm.begin(SSD1306_SWITCHCAPVCC, 0x3C);

  scherm.setTextColor(SSD1306_WHITE);

}

void loop() {

  long afstand = meetAfstand();

  toonAfstand(afstand);

  if (afstand < 5) {

    digitalWrite(ZOEMER_PIN, HIGH);   // Heel dichtbij: één lange toon

    delay(100);

  } else if (afstand < 50) {

    digitalWrite(ZOEMER_PIN, HIGH);   // Korte piep

    delay(50);

    digitalWrite(ZOEMER_PIN, LOW);

    delay(afstand * 10);              // Dichterbij = sneller piepen

  } else {

    digitalWrite(ZOEMER_PIN, LOW);

    delay(100);

  }

}

http://app/nl/starter-kits/uno/tutorial/kit/hc-sr04
http://app/nl/starter-kits/uno/tutorial/kit/active-buzzer-module
http://app/nl/starter-kits/uno/tutorial/kit/oled-display-091-inch

